

## excel yourself

Autoren: Ralf Sowa, Christian Hapke

Beachten Sie unsere [Hinweise](#) und [Nutzungsbedingungen](#). Vorgestellte Musterlösungen basieren auf MS-Excel® 2003; sie gelten ausschließlich für aufgezeigte Beispieldaten. Bitte melden Sie uns etwaige Fehler in unseren Informationen – Ihr Feedback ist willkommen: [urs.toolbox@urs-beratung.de](mailto:urs.toolbox@urs-beratung.de)

Eine Übersicht zu unseren Excel-Informationen finden Sie hier: [www.urs-beratung.de/toolbox.htm](http://www.urs-beratung.de/toolbox.htm)

### Excel - Thema: Finanzmathematik

Funktionen: BW, IKV, KAPZ, NBW, RMZ, ZINS, ZINSZ, ZW, ZZR

## Übersicht

	A	B	C	D	E	F
1	<b>Basisdaten</b>					
2	Barwert		10.000,00 €			
3	Laufzeit (Jahre)		4			
4	Zinssatz (p.a.)		5%			
5	Endwert/Schlusszahlg		-2.000,00 €			
6	Raten am Ende der Periode		0			
7						
8		Annuität	-2.356,09 €	=RMZ(C4;C3;C2;C5;C6)		
9		Barwert	10.000,00 €	=BW(C4;C3;C8;C5;C6)		
10		Laufzeit (Jahre)	4	=ZZR(C4;C8;C2;C5;C6)		
11		Zinssatz p.a.	5%	=ZINS(C3;C8;C2;C5;C6)		
12		Endwert	-2.000,00 €	=ZW(C4;C3;C8;C2;C6)		
13						
14						
15	Periode	+	-		=	
		AB	Tilgung		EB	
16	1	10.000,00 €	-1.856,09 €	=KAPZ(C\$4;A16;C\$3;C\$2;C\$5;C\$6)	8.143,91 €	=B16+C16
17	2	8.143,91 €	-1.948,90 €	=KAPZ(C\$4;A17;C\$3;C\$2;C\$5;C\$6)	6.195,01 €	=B17+C17
18	3	6.195,01 €	-2.046,34 €	=KAPZ(C\$4;A18;C\$3;C\$2;C\$5;C\$6)	4.148,66 €	=B18+C18
19	4	4.148,66 €	-2.148,66 €	=KAPZ(C\$4;A19;C\$3;C\$2;C\$5;C\$6)	2.000,00 €	=B19+C19
20	Schlussz.	2.000,00 €	-2.000,00 €	=C5	0,00 €	=B20+C20
21	Σ		-10.000,00 €	=SUMME(C16:C20)		
22						
23						
24	Periode	+	+		=	
		Tilgung	Zins		Annuität	
25	1	-1.856,09 €	-500,00 €	=ZINSZ(C\$4;A25;C\$3;C\$2;C\$5;C\$6)	-2.356,09 €	=B25+C25
26	2	-1.948,90 €	-407,20 €	=ZINSZ(C\$4;A26;C\$3;C\$2;C\$5;C\$6)	-2.356,09 €	=B26+C26
27	3	-2.046,34 €	-309,75 €	=ZINSZ(C\$4;A27;C\$3;C\$2;C\$5;C\$6)	-2.356,09 €	=B27+C27
28	4	-2.148,66 €	-207,43 €	=ZINSZ(C\$4;A28;C\$3;C\$2;C\$5;C\$6)	-2.356,09 €	=B28+C28
29	Schlussz.	-2.000,00 €			-2.000,00 €	=B29+C29
30	Σ	-10.000,00 €	-1.424,38 €	=SUMME(C25:C29)	-11.424,38 €	
31						
32		Barwert	10.000,00 €	=-NBW(C4;C33;C36)		
33		Zahlg 1. Jahr	-2.356,09 €	=RMZ(C\$4;C\$3;C\$2;C\$5;C\$6)		
34		Zahlg 2. Jahr	-2.356,09 €	=RMZ(C\$4;C\$3;C\$2;C\$5;C\$6)		
35		Zahlg 3. Jahr	-2.356,09 €	=RMZ(C\$4;C\$3;C\$2;C\$5;C\$6)		
36		Zahlg 4. Jahr	-4.356,09 €	=RMZ(C\$4;C\$3;C\$2;C\$5;C\$6)+C5		
37						
38		Zinssatz (p.a.)	5%	=IKV(C32;C36)		

## Funktion oder Argument

BW, ZINS, ZZR, RMZ und ZW haben zunächst eine Bedeutung als **Funktion**. Als solche beinhaltet ihr Ergebnis eine *Aussage*.

- BW ist der *Barwert* von irgendetwas.
- ZINS ist der zugrunde liegende Zinssatz.
- ZZR ist die Anzahl der Zahlungen.
- RMZ ist die Höhe der einzelnen Zahlungen (Raten).
- ZW ist der Zielwert (Endwert).

Daneben sind BW, ZINS, ZZR, RMZ und ZW aber auch **Argumente** zur Berechnung der jeweils anderen Funktionen. Und die *Funktionen* können gleichermaßen bei der Berechnung eines Darlehens, einer Geldanlage oder einer sonstigen (betrieblichen) Investition eingesetzt werden. Als *Argument* treffen BW, ZINS, ZZR, RMZ und ZW – je nach **Zweck der Berechnung** (Darlehen, Geldanlage, Investition) – eine etwas andere *Aussage*.

- Bei einem Darlehen ist BW der aufgenommene Kreditbetrag.
- Bei einer Geldanlage wäre BW ein Startguthaben.
- Bei einer Investition entspricht BW z. B. den Anschaffungskosten einer Maschine.

Wir wollen versuchen, all diesen Aspekten gerecht zu werden:

## BW, ZINS, ZZR, RMZ, ZW

Die Funktionen BW, ZINS, ZZR, RMZ, ZW korrelieren miteinander: Zur Berechnung von BW (Barwert) sind die Angaben der übrigen Werte ZINS, ZZR, RMZ und ZW erforderlich (gleiches gilt für die Berechnung der übrigen Werte):



**Funktionsargumente**

BW

Zins  = Zahl

Zzr  = Zahl

Rmz  = Zahl

Zw  = Zahl

F  = Zahl

=

Gibt den Barwert einer Investition zurück: den heutigen Gesamtwert einer Reihe zukünftiger Zahlungen.

Auf den Wert für „F“ (siehe obige Grafik) gehen wir im Anschluss an die Erläuterungen der fünf Funktionen ein – siehe „F – Zahlungszeitpunkt der Raten“.

**BW**

BW als Funktion / ist gesuchter Wert	BW ist der auf den Zeitpunkt $t_0$ abgezinste Wert (= Barwert) einer Zahlungsreihe Die Abzinsung erfolgt zu gleichbleibendem Zinssatz (ZINS), die Zahlungsreihe besteht aus gegebenen Anzahl (ZZR) gleicher Raten (RMZ) sowie eines Endwerts (ZW). positive Zahl bedeutet Mittelzufluss (ggf. Investition <i>lohnend</i> ) negative Zahl bedeutet Mittelabfluss (ggf. Investition <i>nicht lohnend</i> )		
BW als Argument zur Berechnung anderer Funktion	Darlehen	Geldanlage	Investition
	BW ist der Darlehensbetrag positive Zahl, da Mittelzufluss	BW ist ein Startkapital vor Beginn ratierlicher Einzahlungen negative Zahl, da Mittelabfluss	BW ist der Betrag der Investition negative Zahl, da Mittelabfluss

**ZINS**

ZINS als Funktion / ist gesuchter Wert	ZINS ist der Zinssatz Zu diesem Zinssatz erfolgt die Abzinsung einer Zahlungsreihe auf den Barwert (BW) – die Zahlungsreihe besteht aus Anzahl (ZZR) gleicher Raten (RMZ) sowie des Endwerts (ZW).		
ZINS als Argument zur Berechnung anderer Funktion	Darlehen	Geldanlage	Investition
	ZINS ist der Zinssatz, in dessen Höhe die Zinsen auf gewährtes Darlehen zu zahlen sind	ZINS ist der Zinssatz, in dessen Höhe Zinsen auf ein Start-Guthaben und ratierliche Einzahlungen geleistet werden	ZINS ist der zugrunde gelegte Zinssatz

ZINS ist als **6%** oder **0,06** einzugeben.

In der Regel bezeichnet ein Zinssatz die pro Jahr zu leistenden Zinsen (p.a. = per annum).

ZINS korreliert mit der Anzahl der Zahlungsperioden (ZZR) und ist entsprechend *umzurechnen*.  
 Beispiel: Beträgt der Zinssatz 6% p.a. ist ZINS

- bei monatlichen Raten (RMZ) als **=6%/12** oder **0,5%**
- für quartalsweise Raten (RMZ) als **=6%/4** bzw. 1,5%.

einzugeben.

**ZZR**

ZZR als Funktion / ist gesuchter Wert	ZZR ist die Anzahl der Zahlungsperioden. Die Anzahl von Zahlungen gleicher Raten (RMZ) ist bei gleichbleibendem Zinssatz (ZINS) erforderlich, damit aus dem Barwert (BW) der Zielwert (ZW) entsteht.		
ZZR als Argument zur Berechnung anderer Funktion	Darlehen	Geldanlage	Investition
	ZZR ist die Anzahl der Raten (Annuitäten-Raten)	ZZR ist die Anzahl der (Spar-) Raten	ZZR ist die Anzahl der periodischen Raten (RMZ = erwartete Erträge je Periode)

ZZR korreliert mit ZINS (siehe oben) und ist entsprechend *umzurechnen*. Beträgt die Laufzeit 5 Jahre ist ZZR

- bei monatlichen Zahlungen als  $=5*12$  oder **60**
- bei quartalsweisen Zahlungen als  $=5*4$  oder **20**

**RMZ**

RMZ als Funktion / ist gesuchter Wert	RMZ ist die gleichbleibende Rate. Die Rate (RMZ) ist bei gleichbleibendem Zinssatz (ZINS) so häufig (ZZR) zu leisten, damit aus dem Barwert (BW) der Zielwert (ZW) entsteht.		
RMZ als Argument zur Berechnung anderer Funktion	Darlehen	Geldanlage	Investition
	RMZ ist die Rate oder Annuität – sie besteht aus Zins und Tilgung  Ist BW als positive Zahl (Mittelzufluss) erfasst, liefert RMZ eine negative Zahl (Mittelabfluss)	RMZ ist die gleichbleibende (Spar-) Rate  Ist BW als negative Zahl (Mittelabfluss) erfasst, liefert RMZ eine positive Zahl (Mittelzufluss)	RMZ ist der gleichbleibende (durchschnittliche) Ertrag je Periode

**ZW**

ZW als Funktion / ist gesuchter Wert	ZW ist der Ziel- oder Endwert. Gleichbleibende Raten (RMZ), die in vorgegebener Anzahl (ZZR) geleistet werden, und ein Startwert (BW) ergeben bei einem festen Zinssatzes (ZINS) den Endwert (ZW).		
ZW als Argument zur Berechnung anderer Funktion	Darlehen	Geldanlage	Investition
	ZW ist die verbleibende Restschuld nach allen betrachteten Zahlungen. ZW ist Null, wenn das Darlehen am Ende aller Zahlungen vollständig getilgt sein soll.	ZW ist der End- oder Zielwert	ZW ist der End- oder Zielwert, das kann ein Schrottwert sein.

Erfolgt keine Angabe des ZW, gilt dies als Wert „Null“.

## F – Zahlungszeitpunkt der Raten

Zur Berechnung einer der Funktionen BW, ZW, RMZ, ZINS, ZZR ist – jeweils neben den vier anderen als Argumente in der Formel anzugeben Werten – auch das Kennzeichen „F“ zu definieren. Mit „F“ wird definiert, ob die Raten am Ende (F = 0 oder keine Angabe) oder am Anfang (F = 1) der Perioden gezahlt werden. Nochmals: Es geht hier um den **Zeitpunkt der Zahlungen**, ob also die Raten (RMZ) zu Beginn oder am Ende einer jeden Periode (ZZR) gezahlt werden.

Welche Bedeutung das hat, schauen wir uns an folgendem Beispiel an: Wir berechnen den Wert Zielwert ZW (Spalte F). Die übrigen Werte (BW, RMZ, ZINS und ZZR) sind Konstanten. Zeilen 3 und 4 unterscheiden sich in dem Kennzeichen „F“ (Spalte E).

Die Frage zu unserer Berechnung lautet: Über welches Kapital verfügen wir am Ende der Betrachtung, wenn wir einmalig (ZZR = 1) einen Betrag / eine Rate (RMZ = € 100) zu bekanntem Zinssatz (ZINS = 10%) anlegen? Der Vollständigkeit halber: ein Start- / Barwert (BW = 0) ist nicht vorhanden (Null).

Wir erhalten jährlich 10% Zinsen auf das Guthaben auf unserem *Sparbuch*. Das können wir im Kopf ausrechnen: 10% von € 100 sind € 10 – nach einem Jahr sind € 110 auf dem Sparbuch.

	A	B	C	D	E	F	G
1	Startbetrag	Raten	Zinssatz	Anzahl Zahlungen	Zeitpunkt	Endwert	
2	BW	RMZ	ZINS	ZZR	F	ZW	
3	0,00 €	-100,00 €	10%	1	1	110,00 €	=ZW(C3;D3;B3;A3;E3)
4	0,00 €	-100,00 €	10%	1	0	100,00 €	=ZW(C4;D4;B4;A4;E4)

In Zelle F3 ist das gelungen, in F4 nicht. Warum?

In Zelle E3 haben wir definiert, dass die *Zahlung* am **Anfang** der Periode erfolgt (F = 1): Wir zahlen also zu Beginn der Periode den Betrag auf unser Sparbuch ein. Unser Ersparnes liegt 1 Jahr bei dem großzügigen Zinszahler. Nach einem Jahr erhalten wir Zinsen. Am Ende verfügen wir über € 110.

In Zelle E4 haben wir definiert, dass die *Zahlung* am **Ende** der Periode erfolgt (F = 0). Wir zahlen erst am Ende der Periode den Betrag auf unser Sparbuch ein. Während des ersten Jahres weist unser Sparbuch eine farblose Null aus. 10% Zins von Null sind Null. Unser Endwert (ZW) zeigt folglich unsere Zahlung ohne jeden Zins.

Bei dem Kennzeichen F denken wir zwar vor- und nachschüssige Zinsen. Die Funktionen sind dafür aber nicht wirklich geeignet – sie gehen von schlichten Zahlungsreihen aus (Zahlung der Raten, nicht der Zinsen).

## Periodische Zahlungsreihen

Die Funktionen BW, ZINS, ZZR, RMZ und ZW eignen sich (nur) zur Berechnung periodischer Zahlungsreihen.

Mit ZW berechnen Sie beispielsweise die Endsumme, die Sie aus einer monatlichen Sparrate von € 100 über 10 Jahre bei einem angenommenen Zinssatz von 6% erwarten.

=ZW(6%/12;120;-100;0;0)  
16.387,93 €

Die Funktionen BW und ZW wenden Sie für den Zeitraum an, während dessen Zahlungen (gleichmäßig) erfolgen. Für vor der ersten oder nach der letzten Zahlung liegende Zeiträume ist auf üblichem Wege mittels Zinsformeln aufzuzinsen bzw. abzuzinsen.

Soll das während 10-jähriger Ansparzeit angesammelte Kapital noch weitere 3 Jahre zu 6% angelegt bleiben, bevor Sie darüber verfügen, zinsen Sie das mit ZW berechnete Kapital auf:

=16387,93\*1,06^3  
=ZW(6%/12;120;-100;0;0)\*1,06^3

## Beispiele

### Fonds-Sparen

Hier berechnen wir den Zielwert ZW für eine regelmäßige Sparrate mit Einmalzahlung zu Beginn.

	A	B	C	D	E
1	Sparrate	-100,00 € monatlich			
2	Laufzeit	10 Jahre			
3	Zinssatz p.a.	6%			
4	Einmalzahlung zu Beginn	-1.000,00 €			
5			12 =WENN(C1="monatlich";12;WENN(C1="quartalsweise";4;1))		
6	Zielwert	18.289,27 €	=ZW(B3/B5;B2*B5;B1;B4;1)		

Obige „Sparrate“ und „Einmalzahlung“ sind die Nettobeträge, die tatsächlich in den Fonds eingezahlt werden. Dazu kommt allerdings noch das Agio – nehmen wir es mit 5% an.

Was haben wir denn nun *verdient*?

Wir kennen die Auszahlungen (Startkapital und Raten, jeweils zzgl. der Agio) sowie das Zielkapital:

	A	B	C	D
31	Startkapital	1.000,00 €		
32	erspartes Agio	50,00 €		
33	mtl. Raten	100,00 €		
34	erspartes Agio	5,00 €		
35	Zielkapital	18.289,27 €	=ZW(6%/12;10*12;-100;-1000;1)	
36		5,18%	=ZINS(10*12;-105;-1050;B35;1)*12	
37				
38	davon für Einmalanlage			
39	Startkapital	1.000,00 €		
40	erspartes Agio	50,00 €		
41	Zielkapital	1.819,40 €	=B39*1,005^120	
42		5,51%	=ZINS(10*12;0;-1050;B41;1)*12	
43				
44	davon für mtl. Raten			
45	mtl. Raten	100,00 €		
46	erspartes Agio	5,00 €		
47	Zielkapital	16.469,87 €	=ZW(6%/12;120;-B45;0;1)	
48		5,11%	=ZINS(10*12;-105;0;B47;1)*12	

### Darlehen

	A	B	C
1	Darlehenssumme	10.000,00 €	
2	Laufzeit, Jahre	3	
3	Tilgung	quartalsweise	
4	Zinssatz	6,75%	
5			
6	Annuität pro Quartal	-927,54 €	=RMZ(B4/4;B2*4;B1;0;0)


**Leasingvertrag**

	A	B	C
9	<b>Aus Sicht der Leasinggesellschaft (Leasingraten am Anfang der Periode fällig)</b>		
10			
11	Kaufpreis incl. NK	-50.000,00 €	
12	Laufzeit, Monate	36	
13	Restwert	20.000,00 €	
14	Zinssatz	4%	
15	Leasingrate	949,22 €	=RMZ(B14/12;B12;B11;B13;1)
16			
17	Kaufpreis incl. NK	-50.000,00 €	
18	Anfangszahlung	5.000,00 €	
19	Laufzeit, Monate	36	
20	Restwert	20.000,00 €	
21	Zinssatz	4%	
22	Leasingrate	802,09 €	=RMZ(B21/12;B19;B17+B18;B20;1)
23			
24	<b>Kauf - aus Sicht des Kreditnehmers (Annuität am Ende der Periode)</b>		
25			
26	Kaufpreis incl. NK	50.000,00 €	
27	möglicher Rabatt	-5.000,00 €	
28	Finanzierungsbedarf	45.000,00 €	
29	Laufzeit	36	
30	Restwert	20.000,00 €	
31	Darlehenszins	7,50%	
32	Annuität	-902,66 €	=RMZ(B31/12;36;B28;-B30;0)
33			
34	<b>Vergleich Leasing und Kauf (Abzinsung zu Inflationsrate)</b>		
35			
36	Einmalzahlung	-5.000,00 €	
37	Leasingrate	-802,09 €	
38	Laufzeit, Monate	36	
39	interner Zinssatz	2%	
40	Barwert Leasing	33.050,18 €	=BW(B39/12;B38;B37;0;1)+B36*-1
41	Annuität	-902,66 €	
42	Laufzeit, Monate	36	
43	interner Zinssatz	2%	
44	Barwert Kauf	31.514,46 €	=BW(B43/12;B42;B41;0;0)
45	günstiger ist: Kauf		

## KAPZ, ZINSZ

Auch die Berechnungen von KAPZ und ZINSZ setzen die Angabe der von oben bekannten Argumente BW, ZW, ZINS, ZZR voraus – lediglich RMZ ist entbehrlich. Dafür ist die Angabe einer Zr erforderlich.

Zr bezeichnet die Periode (nur diese eine Periode!), für die KAPZ und ZINSZ ein Ergebnis liefert.



### KAPZ

KAPZ berechnet die Kapitalzahlung, die in **einer** Periode geleistet wird. Für welche Periode KAPZ das Ergebnis liefert, ist als Wert Zr in der Formel zu erfassen. – Bei einem Darlehensvertrag liefert KAPZ den Tilgungsbetrag in der bezeichneten Periode.

### ZINSZ

ZINSZ berechnet den Zinsbetrag, der in **einer** Periode geleistet wird. Für welche Periode ZINSZ das Ergebnis liefert, ist als Wert Zr in der Formel zu erfassen. – Bei einem Darlehensvertrag liefert ZINSZ den Anteil der Zinsen in der bezeichneten Periode.

Mit KAPZ und ZINSZ erfolgt also eine Aufgliederung der einzelnen Raten (RMZ) in den Kapital- und Zinsanteil.

## NBW, IKV

Nicht immer liegen die Basisdaten vollständig vor wie in unseren vorgenannten Beispielen.

	B	C	D
32	Barwert	10.000,00 €	=-NBW(C4;C33:C36)
33	Zahlg 1. Jahr	-2.356,09 €	=RMZ(C\$4;C\$3;C\$2;C\$5;C\$6)
34	Zahlg 2. Jahr	-2.356,09 €	=RMZ(C\$4;C\$3;C\$2;C\$5;C\$6)
35	Zahlg 3. Jahr	-2.356,09 €	=RMZ(C\$4;C\$3;C\$2;C\$5;C\$6)
36	Zahlg 4. Jahr	-4.356,09 €	=RMZ(C\$4;C\$3;C\$2;C\$5;C\$6)+C5
37			
38	Zinssatz (p.a.)	5%	=IKV(C32:C36)

### NBW

Liegen ausschließlich Zahlungsreihen (RMZ) in gleichlangen Perioden sowie der Zinssatz (ZINS) vor, liefert NBW den Barwert – NBW ist also die dem BW entsprechende Funktion.

### IKV

Liegen ausschließlich Zahlungsreihen (RMZ) in gleichlangen Perioden sowie der Barwert (BW) vor, liefert IKV den Zinssatz – IKV ist also die dem ZINS entsprechende Funktion.